/*

$$$$$$$\   $$$$$$\   $$$$$$\  $$\      $$\                  
$$  __$$\ $$  __$$\ $$  __$$\ $$$\    $$$ |                 
$$ |  $$ |$$ /  $$ |$$ /  $$ |$$$$\  $$$$ |                 
$$ |  $$ |$$ |  $$ |$$ |  $$ |$$\$$\$$ $$ |                 
$$ |  $$ |$$ |  $$ |$$ |  $$ |$$ \$$$  $$ |                 
$$ |  $$ |$$ |  $$ |$$ |  $$ |$$ |\$  /$$ |                 
$$$$$$$  | $$$$$$  | $$$$$$  |$$ | \_/ $$ |                 
\_______/  \______/  \______/ \__|     \__|                 
                                                            
                                                            
                                                            
 $$$$$$\   $$$$$$\ $$$$$$$$\                                
$$  __$$\ $$  __$$\\__$$  __|                               
$$ /  \__|$$ /  $$ |  $$ |                                  
$$ |      $$$$$$$$ |  $$ |                                  
$$ |      $$  __$$ |  $$ |                                  
$$ |  $$\ $$ |  $$ |  $$ |                                  
\$$$$$$  |$$ |  $$ |  $$ |                                  
 \______/ \__|  \__|  \__|                                  
                                                            
                                                            
                                                            
$$$$$$$\  $$$$$$$$\  $$$$$$\   $$$$$$\  $$\   $$\ $$$$$$$$\ 
$$  __$$\ $$  _____|$$  __$$\ $$  __$$\ $$ |  $$ |$$  _____|
$$ |  $$ |$$ |      $$ /  \__|$$ /  \__|$$ |  $$ |$$ |      
$$$$$$$  |$$$$$\    \$$$$$$\  $$ |      $$ |  $$ |$$$$$\    
$$  __$$< $$  __|    \____$$\ $$ |      $$ |  $$ |$$  __|   
$$ |  $$ |$$ |      $$\   $$ |$$ |  $$\ $$ |  $$ |$$ |      
$$ |  $$ |$$$$$$$$\ \$$$$$$  |\$$$$$$  |\$$$$$$  |$$$$$$$$\ 
\__|  \__|\________| \______/  \______/  \______/ \________|
                                                 
*/

// File @openzeppelin/contracts/token/ERC20/IERC20.sol@v4.0.0

// SPDX-License-Identifier: MIT

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `recipient`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `sender` to `recipient` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address sender, address recipient, uint256 amount) external returns (bool);

    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);
}


// File @openzeppelin/contracts/utils/introspection/IERC165.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC165 standard, as defined in the
 * https://eips.ethereum.org/EIPS/eip-165[EIP].
 *
 * Implementers can declare support of contract interfaces, which can then be
 * queried by others ({ERC165Checker}).
 *
 * For an implementation, see {ERC165}.
 */
interface IERC165 {
    /**
     * @dev Returns true if this contract implements the interface defined by
     * `interfaceId`. See the corresponding
     * https://eips.ethereum.org/EIPS/eip-165#how-interfaces-are-identified[EIP section]
     * to learn more about how these ids are created.
     *
     * This function call must use less than 30 000 gas.
     */
    function supportsInterface(bytes4 interfaceId) external view returns (bool);
}


// File @openzeppelin/contracts/token/ERC721/IERC721.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @dev Required interface of an ERC721 compliant contract.
 */
interface IERC721 is IERC165 {
    /**
     * @dev Emitted when `tokenId` token is transferred from `from` to `to`.
     */
    event Transfer(address indexed from, address indexed to, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables `approved` to manage the `tokenId` token.
     */
    event Approval(address indexed owner, address indexed approved, uint256 indexed tokenId);

    /**
     * @dev Emitted when `owner` enables or disables (`approved`) `operator` to manage all of its assets.
     */
    event ApprovalForAll(address indexed owner, address indexed operator, bool approved);

    /**
     * @dev Returns the number of tokens in ``owner``'s account.
     */
    function balanceOf(address owner) external view returns (uint256 balance);

    /**
     * @dev Returns the owner of the `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function ownerOf(uint256 tokenId) external view returns (address owner);

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If the caller is not `from`, it must be have been allowed to move this token by either {approve} or {setApprovalForAll}.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Transfers `tokenId` token from `from` to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {safeTransferFrom} whenever possible.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 tokenId) external;

    /**
     * @dev Gives permission to `to` to transfer `tokenId` token to another account.
     * The approval is cleared when the token is transferred.
     *
     * Only a single account can be approved at a time, so approving the zero address clears previous approvals.
     *
     * Requirements:
     *
     * - The caller must own the token or be an approved operator.
     * - `tokenId` must exist.
     *
     * Emits an {Approval} event.
     */
    function approve(address to, uint256 tokenId) external;

    /**
     * @dev Returns the account approved for `tokenId` token.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function getApproved(uint256 tokenId) external view returns (address operator);

    /**
     * @dev Approve or remove `operator` as an operator for the caller.
     * Operators can call {transferFrom} or {safeTransferFrom} for any token owned by the caller.
     *
     * Requirements:
     *
     * - The `operator` cannot be the caller.
     *
     * Emits an {ApprovalForAll} event.
     */
    function setApprovalForAll(address operator, bool _approved) external;

    /**
     * @dev Returns if the `operator` is allowed to manage all of the assets of `owner`.
     *
     * See {setApprovalForAll}
     */
    function isApprovedForAll(address owner, address operator) external view returns (bool);

    /**
      * @dev Safely transfers `tokenId` token from `from` to `to`.
      *
      * Requirements:
      *
      * - `from` cannot be the zero address.
      * - `to` cannot be the zero address.
      * - `tokenId` token must exist and be owned by `from`.
      * - If the caller is not `from`, it must be approved to move this token by either {approve} or {setApprovalForAll}.
      * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
      *
      * Emits a {Transfer} event.
      */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes calldata data) external;
}


// File @openzeppelin/contracts/token/ERC721/IERC721Receiver.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @title ERC721 token receiver interface
 * @dev Interface for any contract that wants to support safeTransfers
 * from ERC721 asset contracts.
 */
interface IERC721Receiver {
    /**
     * @dev Whenever an {IERC721} `tokenId` token is transferred to this contract via {IERC721-safeTransferFrom}
     * by `operator` from `from`, this function is called.
     *
     * It must return its Solidity selector to confirm the token transfer.
     * If any other value is returned or the interface is not implemented by the recipient, the transfer will be reverted.
     *
     * The selector can be obtained in Solidity with `IERC721.onERC721Received.selector`.
     */
    function onERC721Received(address operator, address from, uint256 tokenId, bytes calldata data) external returns (bytes4);
}


// File @openzeppelin/contracts/token/ERC721/extensions/IERC721Metadata.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @title ERC-721 Non-Fungible Token Standard, optional metadata extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Metadata is IERC721 {

    /**
     * @dev Returns the token collection name.
     */
    function name() external view returns (string memory);

    /**
     * @dev Returns the token collection symbol.
     */
    function symbol() external view returns (string memory);

    /**
     * @dev Returns the Uniform Resource Identifier (URI) for `tokenId` token.
     */
    function tokenURI(uint256 tokenId) external view returns (string memory);
}


// File @openzeppelin/contracts/token/ERC721/extensions/IERC721Enumerable.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @title ERC-721 Non-Fungible Token Standard, optional enumeration extension
 * @dev See https://eips.ethereum.org/EIPS/eip-721
 */
interface IERC721Enumerable is IERC721 {

    /**
     * @dev Returns the total amount of tokens stored by the contract.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns a token ID owned by `owner` at a given `index` of its token list.
     * Use along with {balanceOf} to enumerate all of ``owner``'s tokens.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) external view returns (uint256 tokenId);

    /**
     * @dev Returns a token ID at a given `index` of all the tokens stored by the contract.
     * Use along with {totalSupply} to enumerate all tokens.
     */
    function tokenByIndex(uint256 index) external view returns (uint256);
}


// File @openzeppelin/contracts/utils/Address.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @dev Collection of functions related to the address type
 */
library Address {
    /**
     * @dev Returns true if `account` is a contract.
     *
     * [IMPORTANT]
     * ====
     * It is unsafe to assume that an address for which this function returns
     * false is an externally-owned account (EOA) and not a contract.
     *
     * Among others, `isContract` will return false for the following
     * types of addresses:
     *
     *  - an externally-owned account
     *  - a contract in construction
     *  - an address where a contract will be created
     *  - an address where a contract lived, but was destroyed
     * ====
     */
    function isContract(address account) internal view returns (bool) {
        // This method relies on extcodesize, which returns 0 for contracts in
        // construction, since the code is only stored at the end of the
        // constructor execution.

        uint256 size;
        // solhint-disable-next-line no-inline-assembly
        assembly { size := extcodesize(account) }
        return size > 0;
    }

    /**
     * @dev Replacement for Solidity's `transfer`: sends `amount` wei to
     * `recipient`, forwarding all available gas and reverting on errors.
     *
     * https://eips.ethereum.org/EIPS/eip-1884[EIP1884] increases the gas cost
     * of certain opcodes, possibly making contracts go over the 2300 gas limit
     * imposed by `transfer`, making them unable to receive funds via
     * `transfer`. {sendValue} removes this limitation.
     *
     * https://diligence.consensys.net/posts/2019/09/stop-using-soliditys-transfer-now/[Learn more].
     *
     * IMPORTANT: because control is transferred to `recipient`, care must be
     * taken to not create reentrancy vulnerabilities. Consider using
     * {ReentrancyGuard} or the
     * https://solidity.readthedocs.io/en/v0.5.11/security-considerations.html#use-the-checks-effects-interactions-pattern[checks-effects-interactions pattern].
     */
    function sendValue(address payable recipient, uint256 amount) internal {
        require(address(this).balance >= amount, "Address: insufficient balance");

        // solhint-disable-next-line avoid-low-level-calls, avoid-call-value
        (bool success, ) = recipient.call{ value: amount }("");
        require(success, "Address: unable to send value, recipient may have reverted");
    }

    /**
     * @dev Performs a Solidity function call using a low level `call`. A
     * plain`call` is an unsafe replacement for a function call: use this
     * function instead.
     *
     * If `target` reverts with a revert reason, it is bubbled up by this
     * function (like regular Solidity function calls).
     *
     * Returns the raw returned data. To convert to the expected return value,
     * use https://solidity.readthedocs.io/en/latest/units-and-global-variables.html?highlight=abi.decode#abi-encoding-and-decoding-functions[`abi.decode`].
     *
     * Requirements:
     *
     * - `target` must be a contract.
     * - calling `target` with `data` must not revert.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data) internal returns (bytes memory) {
      return functionCall(target, data, "Address: low-level call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`], but with
     * `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        return functionCallWithValue(target, data, 0, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but also transferring `value` wei to `target`.
     *
     * Requirements:
     *
     * - the calling contract must have an ETH balance of at least `value`.
     * - the called Solidity function must be `payable`.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value) internal returns (bytes memory) {
        return functionCallWithValue(target, data, value, "Address: low-level call with value failed");
    }

    /**
     * @dev Same as {xref-Address-functionCallWithValue-address-bytes-uint256-}[`functionCallWithValue`], but
     * with `errorMessage` as a fallback revert reason when `target` reverts.
     *
     * _Available since v3.1._
     */
    function functionCallWithValue(address target, bytes memory data, uint256 value, string memory errorMessage) internal returns (bytes memory) {
        require(address(this).balance >= value, "Address: insufficient balance for call");
        require(isContract(target), "Address: call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.call{ value: value }(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data) internal view returns (bytes memory) {
        return functionStaticCall(target, data, "Address: low-level static call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a static call.
     *
     * _Available since v3.3._
     */
    function functionStaticCall(address target, bytes memory data, string memory errorMessage) internal view returns (bytes memory) {
        require(isContract(target), "Address: static call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.staticcall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data) internal returns (bytes memory) {
        return functionDelegateCall(target, data, "Address: low-level delegate call failed");
    }

    /**
     * @dev Same as {xref-Address-functionCall-address-bytes-string-}[`functionCall`],
     * but performing a delegate call.
     *
     * _Available since v3.4._
     */
    function functionDelegateCall(address target, bytes memory data, string memory errorMessage) internal returns (bytes memory) {
        require(isContract(target), "Address: delegate call to non-contract");

        // solhint-disable-next-line avoid-low-level-calls
        (bool success, bytes memory returndata) = target.delegatecall(data);
        return _verifyCallResult(success, returndata, errorMessage);
    }

    function _verifyCallResult(bool success, bytes memory returndata, string memory errorMessage) private pure returns(bytes memory) {
        if (success) {
            return returndata;
        } else {
            // Look for revert reason and bubble it up if present
            if (returndata.length > 0) {
                // The easiest way to bubble the revert reason is using memory via assembly

                // solhint-disable-next-line no-inline-assembly
                assembly {
                    let returndata_size := mload(returndata)
                    revert(add(32, returndata), returndata_size)
                }
            } else {
                revert(errorMessage);
            }
        }
    }
}


// File @openzeppelin/contracts/utils/Context.sol@v4.0.0



pragma solidity ^0.8.0;

/*
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        this; // silence state mutability warning without generating bytecode - see https://github.com/ethereum/solidity/issues/2691
        return msg.data;
    }
}


// File @openzeppelin/contracts/utils/Strings.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @dev String operations.
 */
library Strings {
    bytes16 private constant alphabet = "0123456789abcdef";

    /**
     * @dev Converts a `uint256` to its ASCII `string` decimal representation.
     */
    function toString(uint256 value) internal pure returns (string memory) {
        // Inspired by OraclizeAPI's implementation - MIT licence
        // https://github.com/oraclize/ethereum-api/blob/b42146b063c7d6ee1358846c198246239e9360e8/oraclizeAPI_0.4.25.sol

        if (value == 0) {
            return "0";
        }
        uint256 temp = value;
        uint256 digits;
        while (temp != 0) {
            digits++;
            temp /= 10;
        }
        bytes memory buffer = new bytes(digits);
        while (value != 0) {
            digits -= 1;
            buffer[digits] = bytes1(uint8(48 + uint256(value % 10)));
            value /= 10;
        }
        return string(buffer);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation.
     */
    function toHexString(uint256 value) internal pure returns (string memory) {
        if (value == 0) {
            return "0x00";
        }
        uint256 temp = value;
        uint256 length = 0;
        while (temp != 0) {
            length++;
            temp >>= 8;
        }
        return toHexString(value, length);
    }

    /**
     * @dev Converts a `uint256` to its ASCII `string` hexadecimal representation with fixed length.
     */
    function toHexString(uint256 value, uint256 length) internal pure returns (string memory) {
        bytes memory buffer = new bytes(2 * length + 2);
        buffer[0] = "0";
        buffer[1] = "x";
        for (uint256 i = 2 * length + 1; i > 1; --i) {
            buffer[i] = alphabet[value & 0xf];
            value >>= 4;
        }
        require(value == 0, "Strings: hex length insufficient");
        return string(buffer);
    }

}


// File @openzeppelin/contracts/utils/introspection/ERC165.sol@v4.0.0



pragma solidity ^0.8.0;

/**
 * @dev Implementation of the {IERC165} interface.
 *
 * Contracts that want to implement ERC165 should inherit from this contract and override {supportsInterface} to check
 * for the additional interface id that will be supported. For example:
 *
 * ```solidity
 * function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
 *     return interfaceId == type(MyInterface).interfaceId || super.supportsInterface(interfaceId);
 * }
 * ```
 *
 * Alternatively, {ERC165Storage} provides an easier to use but more expensive implementation.
 */
abstract contract ERC165 is IERC165 {
    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override returns (bool) {
        return interfaceId == type(IERC165).interfaceId;
    }
}


// File @openzeppelin/contracts/token/ERC721/ERC721.sol@v4.0.0



pragma solidity ^0.8.0;








/**
 * @dev Implementation of https://eips.ethereum.org/EIPS/eip-721[ERC721] Non-Fungible Token Standard, including
 * the Metadata extension, but not including the Enumerable extension, which is available separately as
 * {ERC721Enumerable}.
 */
contract ERC721 is Context, ERC165, IERC721, IERC721Metadata {
    using Address for address;
    using Strings for uint256;

    // Token name
    string private _name;

    // Token symbol
    string private _symbol;

    // Mapping from token ID to owner address
    mapping (uint256 => address) private _owners;

    // Mapping owner address to token count
    mapping (address => uint256) private _balances;

    // Mapping from token ID to approved address
    mapping (uint256 => address) private _tokenApprovals;

    // Mapping from owner to operator approvals
    mapping (address => mapping (address => bool)) private _operatorApprovals;

    /**
     * @dev Initializes the contract by setting a `name` and a `symbol` to the token collection.
     */
    constructor (string memory name_, string memory symbol_) {
        _name = name_;
        _symbol = symbol_;
    }

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(ERC165, IERC165) returns (bool) {
        return interfaceId == type(IERC721).interfaceId
            || interfaceId == type(IERC721Metadata).interfaceId
            || super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721-balanceOf}.
     */
    function balanceOf(address owner) public view virtual override returns (uint256) {
        require(owner != address(0), "ERC721: balance query for the zero address");
        return _balances[owner];
    }

    /**
     * @dev See {IERC721-ownerOf}.
     */
    function ownerOf(uint256 tokenId) public view virtual override returns (address) {
        address owner = _owners[tokenId];
        require(owner != address(0), "ERC721: owner query for nonexistent token");
        return owner;
    }

    /**
     * @dev See {IERC721Metadata-name}.
     */
    function name() public view virtual override returns (string memory) {
        return _name;
    }

    /**
     * @dev See {IERC721Metadata-symbol}.
     */
    function symbol() public view virtual override returns (string memory) {
        return _symbol;
    }

    /**
     * @dev See {IERC721Metadata-tokenURI}.
     */
    function tokenURI(uint256 tokenId) public view virtual override returns (string memory) {
        require(_exists(tokenId), "ERC721Metadata: URI query for nonexistent token");

        string memory baseURI = _baseURI();
        return bytes(baseURI).length > 0
            ? string(abi.encodePacked(baseURI, tokenId.toString()))
            : '';
    }

    /**
     * @dev Base URI for computing {tokenURI}. Empty by default, can be overriden
     * in child contracts.
     */
    function _baseURI() internal view virtual returns (string memory) {
        return "";
    }

    /**
     * @dev See {IERC721-approve}.
     */
    function approve(address to, uint256 tokenId) public virtual override {
        address owner = ERC721.ownerOf(tokenId);
        require(to != owner, "ERC721: approval to current owner");

        require(_msgSender() == owner || ERC721.isApprovedForAll(owner, _msgSender()),
            "ERC721: approve caller is not owner nor approved for all"
        );

        _approve(to, tokenId);
    }

    /**
     * @dev See {IERC721-getApproved}.
     */
    function getApproved(uint256 tokenId) public view virtual override returns (address) {
        require(_exists(tokenId), "ERC721: approved query for nonexistent token");

        return _tokenApprovals[tokenId];
    }

    /**
     * @dev See {IERC721-setApprovalForAll}.
     */
    function setApprovalForAll(address operator, bool approved) public virtual override {
        require(operator != _msgSender(), "ERC721: approve to caller");

        _operatorApprovals[_msgSender()][operator] = approved;
        emit ApprovalForAll(_msgSender(), operator, approved);
    }

    /**
     * @dev See {IERC721-isApprovedForAll}.
     */
    function isApprovedForAll(address owner, address operator) public view virtual override returns (bool) {
        return _operatorApprovals[owner][operator];
    }

    /**
     * @dev See {IERC721-transferFrom}.
     */
    function transferFrom(address from, address to, uint256 tokenId) public virtual override {
        //solhint-disable-next-line max-line-length
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");

        _transfer(from, to, tokenId);
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId) public virtual override {
        safeTransferFrom(from, to, tokenId, "");
    }

    /**
     * @dev See {IERC721-safeTransferFrom}.
     */
    function safeTransferFrom(address from, address to, uint256 tokenId, bytes memory _data) public virtual override {
        require(_isApprovedOrOwner(_msgSender(), tokenId), "ERC721: transfer caller is not owner nor approved");
        _safeTransfer(from, to, tokenId, _data);
    }

    /**
     * @dev Safely transfers `tokenId` token from `from` to `to`, checking first that contract recipients
     * are aware of the ERC721 protocol to prevent tokens from being forever locked.
     *
     * `_data` is additional data, it has no specified format and it is sent in call to `to`.
     *
     * This internal function is equivalent to {safeTransferFrom}, and can be used to e.g.
     * implement alternative mechanisms to perform token transfer, such as signature-based.
     *
     * Requirements:
     *
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     * - `tokenId` token must exist and be owned by `from`.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeTransfer(address from, address to, uint256 tokenId, bytes memory _data) internal virtual {
        _transfer(from, to, tokenId);
        require(_checkOnERC721Received(from, to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Returns whether `tokenId` exists.
     *
     * Tokens can be managed by their owner or approved accounts via {approve} or {setApprovalForAll}.
     *
     * Tokens start existing when they are minted (`_mint`),
     * and stop existing when they are burned (`_burn`).
     */
    function _exists(uint256 tokenId) internal view virtual returns (bool) {
        return _owners[tokenId] != address(0);
    }

    /**
     * @dev Returns whether `spender` is allowed to manage `tokenId`.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     */
    function _isApprovedOrOwner(address spender, uint256 tokenId) internal view virtual returns (bool) {
        require(_exists(tokenId), "ERC721: operator query for nonexistent token");
        address owner = ERC721.ownerOf(tokenId);
        return (spender == owner || getApproved(tokenId) == spender || ERC721.isApprovedForAll(owner, spender));
    }

    /**
     * @dev Safely mints `tokenId` and transfers it to `to`.
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - If `to` refers to a smart contract, it must implement {IERC721Receiver-onERC721Received}, which is called upon a safe transfer.
     *
     * Emits a {Transfer} event.
     */
    function _safeMint(address to, uint256 tokenId) internal virtual {
        _safeMint(to, tokenId, "");
    }

    /**
     * @dev Same as {xref-ERC721-_safeMint-address-uint256-}[`_safeMint`], with an additional `data` parameter which is
     * forwarded in {IERC721Receiver-onERC721Received} to contract recipients.
     */
    function _safeMint(address to, uint256 tokenId, bytes memory _data) internal virtual {
        _mint(to, tokenId);
        require(_checkOnERC721Received(address(0), to, tokenId, _data), "ERC721: transfer to non ERC721Receiver implementer");
    }

    /**
     * @dev Mints `tokenId` and transfers it to `to`.
     *
     * WARNING: Usage of this method is discouraged, use {_safeMint} whenever possible
     *
     * Requirements:
     *
     * - `tokenId` must not exist.
     * - `to` cannot be the zero address.
     *
     * Emits a {Transfer} event.
     */
    function _mint(address to, uint256 tokenId) internal virtual {
        require(to != address(0), "ERC721: mint to the zero address");
        require(!_exists(tokenId), "ERC721: token already minted");

        _beforeTokenTransfer(address(0), to, tokenId);

        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(address(0), to, tokenId);
    }

    /**
     * @dev Destroys `tokenId`.
     * The approval is cleared when the token is burned.
     *
     * Requirements:
     *
     * - `tokenId` must exist.
     *
     * Emits a {Transfer} event.
     */
    function _burn(uint256 tokenId) internal virtual {
        address owner = ERC721.ownerOf(tokenId);

        _beforeTokenTransfer(owner, address(0), tokenId);

        // Clear approvals
        _approve(address(0), tokenId);

        _balances[owner] -= 1;
        delete _owners[tokenId];

        emit Transfer(owner, address(0), tokenId);
    }

    /**
     * @dev Transfers `tokenId` from `from` to `to`.
     *  As opposed to {transferFrom}, this imposes no restrictions on msg.sender.
     *
     * Requirements:
     *
     * - `to` cannot be the zero address.
     * - `tokenId` token must be owned by `from`.
     *
     * Emits a {Transfer} event.
     */
    function _transfer(address from, address to, uint256 tokenId) internal virtual {
        require(ERC721.ownerOf(tokenId) == from, "ERC721: transfer of token that is not own");
        require(to != address(0), "ERC721: transfer to the zero address");

        _beforeTokenTransfer(from, to, tokenId);

        // Clear approvals from the previous owner
        _approve(address(0), tokenId);

        _balances[from] -= 1;
        _balances[to] += 1;
        _owners[tokenId] = to;

        emit Transfer(from, to, tokenId);
    }

    /**
     * @dev Approve `to` to operate on `tokenId`
     *
     * Emits a {Approval} event.
     */
    function _approve(address to, uint256 tokenId) internal virtual {
        _tokenApprovals[tokenId] = to;
        emit Approval(ERC721.ownerOf(tokenId), to, tokenId);
    }

    /**
     * @dev Internal function to invoke {IERC721Receiver-onERC721Received} on a target address.
     * The call is not executed if the target address is not a contract.
     *
     * @param from address representing the previous owner of the given token ID
     * @param to target address that will receive the tokens
     * @param tokenId uint256 ID of the token to be transferred
     * @param _data bytes optional data to send along with the call
     * @return bool whether the call correctly returned the expected magic value
     */
    function _checkOnERC721Received(address from, address to, uint256 tokenId, bytes memory _data)
        private returns (bool)
    {
        if (to.isContract()) {
            try IERC721Receiver(to).onERC721Received(_msgSender(), from, tokenId, _data) returns (bytes4 retval) {
                return retval == IERC721Receiver(to).onERC721Received.selector;
            } catch (bytes memory reason) {
                if (reason.length == 0) {
                    revert("ERC721: transfer to non ERC721Receiver implementer");
                } else {
                    // solhint-disable-next-line no-inline-assembly
                    assembly {
                        revert(add(32, reason), mload(reason))
                    }
                }
            }
        } else {
            return true;
        }
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 tokenId) internal virtual { }
}


// File @openzeppelin/contracts/token/ERC721/extensions/ERC721Enumerable.sol@v4.0.0



pragma solidity ^0.8.0;


/**
 * @dev This implements an optional extension of {ERC721} defined in the EIP that adds
 * enumerability of all the token ids in the contract as well as all token ids owned by each
 * account.
 */
abstract contract ERC721Enumerable is ERC721, IERC721Enumerable {
    // Mapping from owner to list of owned token IDs
    mapping(address => mapping(uint256 => uint256)) private _ownedTokens;

    // Mapping from token ID to index of the owner tokens list
    mapping(uint256 => uint256) private _ownedTokensIndex;

    // Array with all token ids, used for enumeration
    uint256[] private _allTokens;

    // Mapping from token id to position in the allTokens array
    mapping(uint256 => uint256) private _allTokensIndex;

    /**
     * @dev See {IERC165-supportsInterface}.
     */
    function supportsInterface(bytes4 interfaceId) public view virtual override(IERC165, ERC721) returns (bool) {
        return interfaceId == type(IERC721Enumerable).interfaceId
            || super.supportsInterface(interfaceId);
    }

    /**
     * @dev See {IERC721Enumerable-tokenOfOwnerByIndex}.
     */
    function tokenOfOwnerByIndex(address owner, uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721.balanceOf(owner), "ERC721Enumerable: owner index out of bounds");
        return _ownedTokens[owner][index];
    }

    /**
     * @dev See {IERC721Enumerable-totalSupply}.
     */
    function totalSupply() public view virtual override returns (uint256) {
        return _allTokens.length;
    }

    /**
     * @dev See {IERC721Enumerable-tokenByIndex}.
     */
    function tokenByIndex(uint256 index) public view virtual override returns (uint256) {
        require(index < ERC721Enumerable.totalSupply(), "ERC721Enumerable: global index out of bounds");
        return _allTokens[index];
    }

    /**
     * @dev Hook that is called before any token transfer. This includes minting
     * and burning.
     *
     * Calling conditions:
     *
     * - When `from` and `to` are both non-zero, ``from``'s `tokenId` will be
     * transferred to `to`.
     * - When `from` is zero, `tokenId` will be minted for `to`.
     * - When `to` is zero, ``from``'s `tokenId` will be burned.
     * - `from` cannot be the zero address.
     * - `to` cannot be the zero address.
     *
     * To learn more about hooks, head to xref:ROOT:extending-contracts.adoc#using-hooks[Using Hooks].
     */
    function _beforeTokenTransfer(address from, address to, uint256 tokenId) internal virtual override {
        super._beforeTokenTransfer(from, to, tokenId);

        if (from == address(0)) {
            _addTokenToAllTokensEnumeration(tokenId);
        } else if (from != to) {
            _removeTokenFromOwnerEnumeration(from, tokenId);
        }
        if (to == address(0)) {
            _removeTokenFromAllTokensEnumeration(tokenId);
        } else if (to != from) {
            _addTokenToOwnerEnumeration(to, tokenId);
        }
    }

    /**
     * @dev Private function to add a token to this extension's ownership-tracking data structures.
     * @param to address representing the new owner of the given token ID
     * @param tokenId uint256 ID of the token to be added to the tokens list of the given address
     */
    function _addTokenToOwnerEnumeration(address to, uint256 tokenId) private {
        uint256 length = ERC721.balanceOf(to);
        _ownedTokens[to][length] = tokenId;
        _ownedTokensIndex[tokenId] = length;
    }

    /**
     * @dev Private function to add a token to this extension's token tracking data structures.
     * @param tokenId uint256 ID of the token to be added to the tokens list
     */
    function _addTokenToAllTokensEnumeration(uint256 tokenId) private {
        _allTokensIndex[tokenId] = _allTokens.length;
        _allTokens.push(tokenId);
    }

    /**
     * @dev Private function to remove a token from this extension's ownership-tracking data structures. Note that
     * while the token is not assigned a new owner, the `_ownedTokensIndex` mapping is _not_ updated: this allows for
     * gas optimizations e.g. when performing a transfer operation (avoiding double writes).
     * This has O(1) time complexity, but alters the order of the _ownedTokens array.
     * @param from address representing the previous owner of the given token ID
     * @param tokenId uint256 ID of the token to be removed from the tokens list of the given address
     */
    function _removeTokenFromOwnerEnumeration(address from, uint256 tokenId) private {
        // To prevent a gap in from's tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = ERC721.balanceOf(from) - 1;
        uint256 tokenIndex = _ownedTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary
        if (tokenIndex != lastTokenIndex) {
            uint256 lastTokenId = _ownedTokens[from][lastTokenIndex];

            _ownedTokens[from][tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
            _ownedTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index
        }

        // This also deletes the contents at the last position of the array
        delete _ownedTokensIndex[tokenId];
        delete _ownedTokens[from][lastTokenIndex];
    }

    /**
     * @dev Private function to remove a token from this extension's token tracking data structures.
     * This has O(1) time complexity, but alters the order of the _allTokens array.
     * @param tokenId uint256 ID of the token to be removed from the tokens list
     */
    function _removeTokenFromAllTokensEnumeration(uint256 tokenId) private {
        // To prevent a gap in the tokens array, we store the last token in the index of the token to delete, and
        // then delete the last slot (swap and pop).

        uint256 lastTokenIndex = _allTokens.length - 1;
        uint256 tokenIndex = _allTokensIndex[tokenId];

        // When the token to delete is the last token, the swap operation is unnecessary. However, since this occurs so
        // rarely (when the last minted token is burnt) that we still do the swap here to avoid the gas cost of adding
        // an 'if' statement (like in _removeTokenFromOwnerEnumeration)
        uint256 lastTokenId = _allTokens[lastTokenIndex];

        _allTokens[tokenIndex] = lastTokenId; // Move the last token to the slot of the to-delete token
        _allTokensIndex[lastTokenId] = tokenIndex; // Update the moved token's index

        // This also deletes the contents at the last position of the array
        delete _allTokensIndex[tokenId];
        _allTokens.pop();
    }
}


// File contracts/ERC721OnOpenSea.sol



pragma solidity ^0.8.0;

interface ProxyRegistry {
    function proxies(address) external view returns (address);
}

/**
 * @dev helper contract to allow gasless OpenSea listings
 */
contract ERC721OnOpenSea is ERC721Enumerable {
    address public proxyRegistryAddress;

    constructor(
        string memory name,
        string memory symbol,
        address registryProxyAddress_
    ) ERC721(name, symbol) {
        proxyRegistryAddress = registryProxyAddress_;
    }

    function isApprovedForAll(address owner, address operator)
        public
        view
        override
        returns (bool)
    {
        ProxyRegistry proxyRegistry = ProxyRegistry(proxyRegistryAddress);

        if (address(proxyRegistry.proxies(owner)) == operator) {
            return true;
        }

        return super.isApprovedForAll(owner, operator);
    }
}


// File contracts/interfaces/IUniswapV2Router02Minimal.sol



pragma solidity ^0.8.0;

interface IUniswapV2Router02Minimal {
    function WETH() external pure returns (address);
    function swapExactETHForTokensSupportingFeeOnTransferTokens(
        uint256 amountOutMin,
        address[] calldata path,
        address to,
        uint256 deadline
    ) external payable;
}


// File contracts/interfaces/IDoomCat.sol



pragma solidity ^0.8.0;

interface IDoomCat {
    function mint(address, uint256) external;
    function swapAndBurn(uint256) external;
}


// File contracts/DoomCatRescue.sol



pragma solidity ^0.8.0;

/**
 * @dev we rescue the cats
 */
contract DoomCatRescue {
    // reference to, basically, self
    IDoomCat private _doomCat;

    // total number of cats that exist
    uint256 public immutable totalCats;

    // how many cats initially existed to be rescued
    uint256 private immutable _remainingCatsInitial;

    // how many cats are left to be rescued
    uint256 public remainingCats;

    // when rescues can begin
    uint256 public immutable rescueStartTime;

    // initial price of rescues
    uint256 public immutable rescuePriceInitial;

    // current price of rescues
    uint256 public rescuePrice;

    // amount of rescue price increases
    uint256 public immutable rescuePriceIncrement;

    // amount of cats to rescue before price increase is triggered
    uint256 public immutable rescueTrancheSize;

    // how many blocks must exist between rescues for a given address
    uint256 public immutable rescueRateLimit;

    // track last block that an address performed a rescue
    mapping(address => uint256) public rescueLastBlock;

    // how many funds have been collected for rescues (resets every tranche)
    uint256 private collectedRescueFunds;

    event Rescue(address indexed rescuer, uint256 tokenId, uint256 price);

    constructor(
        IDoomCat doomCat_,
        uint256[2] memory catDetails_,
        uint256[5] memory rescueDetails_
    ) {
        _doomCat = doomCat_;

        totalCats = catDetails_[0];
        _remainingCatsInitial = catDetails_[1];
        remainingCats = catDetails_[1];

        rescuePriceInitial = rescueDetails_[0];
        rescuePrice = rescueDetails_[0];
        rescuePriceIncrement = rescueDetails_[1];
        rescueTrancheSize = rescueDetails_[2];
        rescueStartTime = rescueDetails_[3];
        rescueRateLimit = rescueDetails_[4];
    }

    /**
     * @dev public function to rescue a cat
     */
    function rescue() public payable {
        // require that rescues have globally started
        require(block.timestamp >= rescueStartTime, "too early");
        
        // require that address has passed their rate limit
        require(
            block.number >= rescueLastBlock[msg.sender] + rescueRateLimit,
            "too soon"
        );

        // require that there are cats left to be rescued
        require(remainingCats > 0, "no cats left");

        // require that the correct amount is paid (at minimum)
        require(msg.value >= rescuePrice, "value too low");

        // update address' last block rate limit tracker
        rescueLastBlock[msg.sender] = block.number;

        // grab the current rescue price (for use after updating `rescuePrice`)
        uint256 currentRescuePrice = rescuePrice;

        // update the amount of funds that have been used for rescues
        collectedRescueFunds += currentRescuePrice;

        // decrement remaining cats
        remainingCats--;

        // if we are at a tranch boundary, OR, there are no remaining cats, then
        // we want to update the price and swap & burn all collected ETH from previous
        // tranche
        if (
            (_remainingCatsInitial - remainingCats) % rescueTrancheSize == 0 ||
            remainingCats == 0
        ) {
            // if there are no more cats, set rescuePrice at 0 for cleanup sake
            if (remainingCats == 0) {
                rescuePrice = 0;
            } else {
                // otherwise, increment rescuePrice by the increment amount
                rescuePrice += rescuePriceIncrement;
            }

            // swap and burn collected funds
            _doomCat.swapAndBurn(collectedRescueFunds);

            // reset collected funds back to 0
            collectedRescueFunds = 0;
        }

        // get the tokenId for the cat to mint, mint it
        uint256 tokenId = totalCats - remainingCats;
        _doomCat.mint(msg.sender, tokenId);

        // if the user overpaid, refund their Ether
        if (msg.value > currentRescuePrice) {
            payable(msg.sender).transfer(msg.value - currentRescuePrice);
        }

        emit Rescue(msg.sender, tokenId, rescuePrice);
    }
}


// File contracts/DoomCatAuction.sol



pragma solidity ^0.8.0;

/**
 * @dev we auction the genesis cats
 */
contract DoomCatAuction {
    // reference to, basically, self
    IDoomCat private _doomCat;

    // number of total genesis cats
    uint256 public immutable genesisCats;

    // start time of the first auction
    uint64 public immutable auctionsStart;

    // how long auctions last
    uint64 public immutable auctionsDuration;

    // time between end of one auction, and beginning of next auction
    uint64 public immutable auctionsDistance;

    // number of total auctions to perform
    uint64 public immutable auctionsCount;

    // number of cats being auctioned off per auction
    uint64 public immutable auctionsCatsPerAuction;

    // minimum bid increase amount
    uint64 public immutable auctionsBidIncrement;

    // track number of bids that have been placed for a cat
    mapping(uint256 => uint256) public bidCount;

    // track the highest bid amount for a cat
    mapping(uint256 => uint256) public highBidAmount;

    // track the owner of the highest bid for a cat
    mapping(uint256 => address) public highBidOwner;

    // track all of the bids (per address) for cats
    mapping(uint256 => mapping(address => uint256)) public bidsByTokenByAddress;

    event Bid(
        address indexed account,
        uint256 catId,
        uint256 amount,
        uint256 bidCount
    );
    event WithdrawLowBid(
        address indexed account,
        uint256 catId,
        uint256 amount
    );
    event ClaimWinningBid(
        address indexed account,
        uint256 catId,
        uint256 amount
    );

    constructor(IDoomCat doomCat_, uint64[6] memory auctionsDetails_) {
        _doomCat = doomCat_;

        genesisCats = auctionsDetails_[3] * auctionsDetails_[4];

        auctionsStart = auctionsDetails_[0];
        auctionsDuration = auctionsDetails_[1];
        auctionsDistance = auctionsDetails_[2];
        auctionsCount = auctionsDetails_[3];
        auctionsCatsPerAuction = auctionsDetails_[4];
        auctionsBidIncrement = auctionsDetails_[5];

        require(auctionsDetails_[0] >= block.timestamp, "too late");
    }

    /**
     * @dev helper function which returns two integers:
     * 1) startId: the ID of a cat which signifies the "lowest" ID of the most recent (or current) auction set
     * 2) endId: the ID of a cat which signifies the "highest" ID of the most recent (or current) auction set
     * if (startId > endId), that means that no auction is currently happening, but exposes
     *   information about the most recently completed auction
     * if (startId <= endId), that means that an auction is currently happening, for that range of IDs
     */
    function auctionsState() public view returns (uint256, uint256) {
        uint64 checkedAuctions = 0;
        uint256 startId = 0;
        uint256 endId = 0;

        // loop through each auction
        while (checkedAuctions <= auctionsCount) {
            // calculate the start time of the current auction iteration
            uint64 auctionStart =
                auctionsStart +
                    (checkedAuctions * (auctionsDuration + auctionsDistance));

            // if the auction starts in the future, we're done with this while loop
            if (auctionStart > block.timestamp) {
                break;
            }
            // otherwise, the auction is currently in progress, or over

            // regardless of in progress or over, we need to calculate the startID
            // which is done by multiplying our auction iteration by number of cats
            // per auction plus 1 because cat ids start at 1
            startId = checkedAuctions * auctionsCatsPerAuction + 1;

            // add duration to start time, compare to current timestamp, to see if
            // we're currently in auction
            if (auctionStart + auctionsDuration > block.timestamp) {
                // if we're in an auction, break the loop and don't update endId
                break;
            } else {
                // if the auction ended in the past, update endId
                endId =
                    checkedAuctions *
                    auctionsCatsPerAuction +
                    auctionsCatsPerAuction;
            }

            // iterate
            checkedAuctions++;
        }

        return (startId, endId);
    }

    /**
     * @dev how many genesis cats are remaining
     */
    function remainingGenesisCats() public view returns (uint256) {
        (, uint256 endId) = auctionsState();
        return genesisCats - endId;
    }

    /**
     * @dev place a bid for a cat, referenced by its index in the current auction
     */
    function bid(uint64 index) public payable {
        (uint256 startId, uint256 endId) = auctionsState();

        // bids can only be placed while in an auction
        require(startId > endId, "not in auction");

        // cats are referenced by their index in the auction
        require(index < auctionsCatsPerAuction, "bad index");
        uint256 catId = startId + index;

        // users may increase their bid by sending the difference of the total amount
        // they want to bid, and their current bid
        uint256 newBid = bidsByTokenByAddress[catId][msg.sender] + msg.value;

        // make sure their new bid covers the bid increment amount
        require(
            newBid >= highBidAmount[catId] + auctionsBidIncrement,
            "not enough"
        );

        // increment the bid count
        bidCount[catId] += 1;

        // set the high bid amount
        highBidAmount[catId] = newBid;

        // set the high bid owner
        highBidOwner[catId] = msg.sender;

        // set the user's bid on this cat (for withdraws, later)
        bidsByTokenByAddress[catId][msg.sender] = newBid;

        emit Bid(msg.sender, catId, newBid, bidCount[catId]);
    }

    /**
     * @dev bids which have been outbid can be "withdrawn" and their Ether returned
     * highest current bid cannot be withdrawn
     */
    function withdrawLowBid(uint256 catId) public {
        // if user is the current highest bid for the cat, no can withdraw
        require(msg.sender != highBidOwner[catId], "can't withdraw high bid");

        // get reference to their bid amount
        uint256 bidAmount = bidsByTokenByAddress[catId][msg.sender];

        // make sure the user actually has funds to withdraw
        require(bidAmount > 0, "nothing to withdraw");

        // reset their amount for this cat to 0
        bidsByTokenByAddress[catId][msg.sender] = 0;

        // send the user their funds
        payable(msg.sender).transfer(bidAmount);

        emit WithdrawLowBid(msg.sender, catId, bidAmount);
    }

    /**
     * @dev when auction is over, highest bid can claim their cat!
     */
    function claimWinningBid(uint256 catId) public {
        // confirm that the cat being claimed belongs to a completed auction
        (, uint256 endId) = auctionsState();
        require(catId <= endId, "cat not claimable");

        // if no bids have been placed on a cat, they can be claimed first-come-first-serve
        if (highBidOwner[catId] != address(0)) {
            // otherwise, require that user is the owner of highest bid
            require(msg.sender == highBidOwner[catId], "not winning bid");
        }

        // mint that cat
        _doomCat.mint(msg.sender, catId);

        // only swap & burn if a bid actually exists
        if (highBidAmount[catId] > 0) {
            _doomCat.swapAndBurn(highBidAmount[catId]);
        }

        emit ClaimWinningBid(msg.sender, catId, highBidAmount[catId]);
    }
}


// File contracts/DoomCat.sol



pragma solidity ^0.8.0;






/**
 * @dev we like the cats
 */
contract DoomCat is IDoomCat, DoomCatRescue, DoomCatAuction, ERC721OnOpenSea {
    string private __baseURI;
    
    // there's an owner, but they can only update the BaseURI
    address public owner;

    // track the cat names here
    mapping(uint256 => string) public catNames;

    // how much does a name change cost
    uint256 public nameChangePrice;

    // reference to the token (HATE) that all funds will be swapped into
    address public immutable HATE;

    // reference to the address where all swapped HATE will be burned to
    address public immutable burn;

    // uniswap router address
    IUniswapV2Router02Minimal public immutable uniswapV2Router;
    
    // events for public functions
    event NameCat(address indexed account, uint256 catId, string name);
    event UpdateBaseURI(string baseURI);
    event RevokeOwner();

    constructor(
        address[4] memory addresses,
        string memory baseURI_,
        uint32 normalCats_,
        uint256[5] memory rescueDetails_,
        uint64[6] memory auctionsDetails_,
        uint256 nameChangePrice_
    )
        ERC721OnOpenSea("DoomCatRescue", "DOOM", addresses[3])
        DoomCatRescue(
            IDoomCat(address(this)),
            [
                uint256(
                    auctionsDetails_[3] * auctionsDetails_[4] + normalCats_
                ),
                normalCats_
            ],
            rescueDetails_
        )
        DoomCatAuction(IDoomCat(address(this)), auctionsDetails_)
    {
        HATE = addresses[0];
        uniswapV2Router = IUniswapV2Router02Minimal(addresses[1]);
        burn = addresses[2];
        __baseURI = baseURI_;
        owner = msg.sender;
        nameChangePrice = nameChangePrice_;
    }

    /**
     * @dev override OZ ERC721 _baseURI() function
     */
    function _baseURI() internal view virtual override returns (string memory) {
        return __baseURI;
    }

    /**
     * @dev we want to be able to update this later, to fully decentralize it
     */
    function updateBaseURI(string calldata baseURI) public {
        require(msg.sender == owner, "not owner");
        __baseURI = baseURI;
        emit UpdateBaseURI(baseURI);
    }

    /**
     * @dev after getting baseURI into it's final form, revoke ownership
     */
    function revokeOwner() public {
        require(msg.sender == owner, "not owner");
        owner = address(0);
        emit RevokeOwner();
    }

    /**
     * @dev mint a new token to recipient with specified id, but only from current contract
     */
    function mint(address to_, uint256 tokenId_) public override {
        require(msg.sender == address(this), "can't mint");
        _mint(to_, tokenId_);
    }

    /**
     * @dev given an amount of Ether, swap all for HATE and send to burn address
     */
    function swapAndBurn(uint256 _amount) public override {
        require(msg.sender == address(this), "can't swap and burn");

        address[] memory path = new address[](2);
        path[0] = uniswapV2Router.WETH();
        path[1] = HATE;

        uniswapV2Router.swapExactETHForTokensSupportingFeeOnTransferTokens{
            value: _amount
        }(0, path, burn, block.timestamp);
    }

    /**
     * @dev owner of a cat can name a cat, once
     */
    function nameCat(uint256 catId, string calldata catName) external {
        require(ownerOf(catId) == msg.sender, "not your cat");

        // can only name a cat if it's not already named
        bytes memory currentName = bytes(catNames[catId]);
        require(currentName.length == 0, "already named");

        catNames[catId] = catName;
        IERC20(HATE).transferFrom(msg.sender, burn, nameChangePrice);

        emit NameCat(msg.sender, catId, catName);
    }
}